

MARVEL SUPER HEROES™

DF
DONALD I. FINE, INC.

COMPUTER FUN

BOOK ONE




ALL PROGRAMS
RUN ON

7

TOP MICROS

20 ACTION-FILLED PROGRAMS

750
C11




**ENTER A
WORLD OF
SUPER COMPUTER
ADVENTURE!**

Join the MARVEL SUPER HEROES™ as they valiantly battle the forces of evil and destruction. Use your trusty computer to run 20 super programs that make you a part of the action. Plug in your computer and get set to load it with codes, computer puzzles, games, graphics, and loads of fun!!!

MARVEL SUPER HEROES™

COMPUTER FUN BOOK ONE



By Richard Guaraldo and Susan M. Zakar
Continuity by Dwight Jon Zimmerman



A Parachute Press Book
Donald I. Fine, Inc.

Acknowledgments

The authors wish to thank the faculty and students of the Four Seasons Elementary School in Gambrills, Maryland, for their support in the preparation and testing of these programs.

Thanks to my wife, Clarena, who has been willing to be a "computer widow" as the material for this book was assembled. She has spent many hours providing inspiration and reviewing my creations. Also to our daughter, Patricia, who has been my quality assurance department (guinea pig). She has personally placed her stamp of approval on my work—RG

Thanks to my husband, Joe, who is a real "computer wizard" and whose love, patience, and gentle instruction helped me to understand the spellbinding magic of computers. He spent many a lonely night while I entered incantations (aka programs) into my microcomputers. Without his support this book would not have been possible. Thanks also to my young son, David, who almost always went to bed on time and slept well, which so greatly increased the time I had to program—SMZ

Parachute Press wishes to thank the creative staff at Marvel Comics for all their help and advice. And special thanks to Michael Z. Hobson, Vice President, Publishing, Marvel Comics Group, without whom this project would not have been possible.

Illustrations copyright © 1984 MARVEL COMICS GROUP, a division of Cadence Industries Corporation. Text copyright © 1984 Parachute Press, Inc.

Spider-Man, Iron Man, Captain America, Thor, Loki, Mr. Fantastic, The Invisible Girl, The Thing, The Human Torch, Doctor Doom, Doctor Octopus, Professor X, The X-Men, Cyclops, Wolverine, Storm, Colossus, Nightcrawler, Ariel, The Avengers, Scarlet Witch, Vision, Doctor Strange, Dormammu, Peter Parker, Daredevil, and the distinctive likenesses thereof are Trademarks of the Marvel Comics Group, a division of Cadence Industries Corporation and are used with permission.

Marvel Super Heroes™ is a trademark of the Marvel Comics Group, a division of Cadence Industries Corporation. Used with permission.

All rights reserved, including the right of reproduction in whole or in part in any form. Published in the United States of America by Donald I. Fine, Inc. and in Canada by Fitzhenry & Whiteside, Ltd.


ISBN: 0-917657-05-5

Manufactured in the United States of America

10 9 8 7 6 5 4 3 2 1

VIC 20 is a trademark of Commodore Computer.

PARACHUTE PRESS
GRATEFULLY DEDICATES
THIS BOOK
TO ***JIM SHOOTER***
MARVEL COMICS EDITOR-IN-CHIEF



ATTENTION, ADVENTURE LOVERS!!

YOUR COMPUTER IS ABOUT TO TURN YOU INTO A SUPER HERO! WITH THE BASIC PROGRAMS IN THE BOOK, YOU WILL FEEL AS IF YOU ARE SCALING WALLS WITH THE AMAZING SPIDER-MAN! YOU WILL HAVE THE SUPER SENSES OF DAREDEVIL. THE INCREDIBLE MIND OF REED RICHARDS, LEADER OF THE FANTASTIC FOUR. THE MAGICAL POWERS OF THOR AND DOCTOR STRANGE. YOU WILL OUTWIT THE VILLAINOUS HOBGOBLIN, TANGLE WITH DOCTOR DOOM AND GO UP AGAINST THE RUTHLESS KINGPIN!

ALL YOU HAVE TO DO IS TYPE THESE PROGRAMS INTO YOUR COMPUTER. THEN ENTER THE MAGIC WORD *RUN*, AND YOU'LL BE OFF INTO A WORLD OF DANGER, EXCITEMENT, HEROISM, AND VILLAINY.

BE CAREFUL. YOU MUST FOLLOW ALL THE INSTRUCTIONS VERY CAREFULLY--OR YOUR CAREER AS A SUPER HERO WILL BE ALL TOO BRIEF. START WITH THE *HOW TO USE THIS BOOK* SECTION ON THE NEXT PAGE. IT HAS THE IMPORTANT INFORMATION YOU REALLY NEED. WHAT ARE YOU WAITING FOR? *THE MARVEL SUPER HEROES™* CHARACTERS AND A WORLD OF COMPUTER FUN ARE WAITING FOR *YOU!*



How to use this book


All the programs in this book have been designed to run in the BASIC programming language on the **IBM PC, PCjr, Commodore 64, VIC 20, Radio Shack Model III, Radio Shack Color Computer (with extended BASIC), Apple II, Apple II+, and Apple IIe, IIC.**

Getting the same BASIC programs to run on many different microcomputers is a difficult task, especially if you want to do anything more than compute and print. Each microcomputer may have a different way of doing things, like clearing the screen, positioning the cursor, or even generating a random number. We wanted to make it as simple as possible. That's why we created the "900-Lines."

Now don't panic! There are not NINE HUNDRED lines. There are less than ten. They have line numbers from 900 to 990 (the Radio Shack III and Color Computer have a line 50, too). Each microcomputer we have listed has its own set of 900-Lines. We've explained them in detail below. The 900-Lines are the only lines that change between computers. All the rest of the BASIC program lines are exactly the same for an IBM PC, an APPLE, COMMODORE 64, VIC 20, Radio Shack III, or Color Computer and PCjr. Just use the ten or so lines for your computer with each program, and PRESTO, it runs! Simple enough? Let's make it even simpler . . .

Most of the 900-Lines don't even change from program to program. That means that you can type them in ONCE, save them just like a BASIC program, and then use them for program after program after program! Here's how to do it:

First, make sure your computer is in BASIC. Then type in the 900-Lines for your microcomputer from the listings. Next, save these lines the same way you would save any BASIC program to tape or disk. We especially like the name 'N900' for this program. Now, BEFORE you type in any of the programs in the book, LOAD the 900-Lines. Type in the rest of the program lines, along with any special changes noted (always lines 970 to 990) and your program is complete. Save the whole new program under its name if you want to load and run it again later.



Here's what all the 900-Lines do:

900	CLEAR SCREEN	This subroutine clears the screen. It does not change from program to program.
910	POSITION CURSOR	This subroutine moves the cursor to a specific place on the screen. The variable VT (Vertical Tab) sets the position up or down on the screen. The variable HT (Horizontal Tab) sets the position across the screen. This line does not change from program to program. VT and HT must be properly set before calling this subroutine.
920	WAIT LOOP	This subroutine is called a WAIT LOOP because all it does is go around in a loop the number of times specified by WT. This subroutine is useful for slowing down programs that run too fast in places. It is the same for all programs and all machines.
930	RANDOM NUMBER GENERATOR	This subroutine returns a RANDOM number from 1 to the value of the variable RX. The number is returned in the variable RD. This line does not change from program to program.
940-950	KEYPRESS	This is the KEYPRESS routine. It detects whether a key has been pressed on the keyboard. If one has been pressed, the character is returned in the string variable KY\$. If no key has been pressed, KY\$ is set to NU\$, or CHR\$(0). These lines do not change from program to program.
960	SETUP	This is the SETUP line. It sets the variables for Screen Height (SH) and Screen Width (SW) and any other variables required by the other 900-Lines. It also makes a call to line 970 where program-specific parameters are set up. This line does not change from program to program.
970-990	PROGRAM SPECIFIC SETUP	These program lines are for setting up variables and parameters which are peculiar for any one program. These lines <i>MAY CHANGE</i> from program to program. They may seed a random number generator, or set the speed for the wait loop or things like that. All three lines are not always used, but they are available just in case. The last line actually used is the one that ends with a RETURN statement.

**DON'T FORGET
YOUR 900-LINES!
THEY CUSTOMIZE
THE PROGRAMS
FOR YOUR
COMPUTER.**

After you have loaded the 900-Lines, all you have to do is input the BASIC program in the text. Be very careful to type the listing just as it appears. Even an extra space or a comma can prevent the program from running correctly. Also remember to use ALL CAPITAL LETTERS when you are inputting data or hitting a key to play a game.

Some of the programs have been designed to give you the challenge of debugging them. That means that you must change something in the program so that it will run correctly. The instructions will tell you when you've come across one of these programs. And if you get stumped as to how to fix it, there's help in the Super Hint Section in the back of the book.

Have fun!



Use the following lines for the COMMODORE 64

```
900 PRINT CHR$(147):RETURN
910 POKE XT,HT-1:POKE YT,VT-1:POKE FG,0:SYS PL:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RX*RND(1)+1):RETURN
940 GET KY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 XT=782:YT=781:FG=783:PL=65520:SW=40:SH=24:NU$=CHR$(0):
    GOSUB 970:RETURN
970 RD=RND(-TI):RETURN
```

Use the following lines for the COMMODORE VIC 20

```
900 PRINT CHR$(147):RETURN
910 POKE XT,HT-1:POKE YT,VT-1:POKE FG,0:SYS PL:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RX*RND(1)+1):RETURN
940 GET KY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 XT=782:YT=781:FG=783:PL=65520:SW=22:SH=22:NU$=CHR$(0):GOSUB
    970:RETURN
970 RD=RND(-TI):RETURN
```

Use the following lines for the RADIO SHACK COLOR COMPUTER

```
50 CLEAR 2000
900 CLS:RETURN
910 HZ=INT(HT-1+(VT-1)*SW+0.5):PRINT@ HZ,"":RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RND(RX)):RETURN
940 KY$=INKEY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 NU$=CHR$(0):SW=32:SH=16:GOSUB 970:RETURN
970 RETURN
```

Use the following lines for the RADIO SHACK MODEL III

```
500 CLEAR 2000
900 CLS:RETURN
910 HZ=INT(HT-1+(VT-1)*SW+0.5):PRINT@ HZ,"";:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RND(RX)):RETURN
940 KY$=INKEY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 NU$=CHR$(0):SW=64:SH=16:GOSUB 970:RETURN
970 RETURN
```

Use the following lines for the IBM PC, and IBM PCjr

```
900 CLS:RETURN
910 LOCATE VT,HT:RETURN
920 FOR WS=1 TO WT:NEXT:RETURN
930 RD=INT(RX*RND(1)+1):RETURN
940 KY$=INKEY$:IF KY$="" THEN KY$=NU$
950 RETURN
960 SW=40:SH=24:NU$=CHR$(0):GOSUB 970:RETURN
970 RANDOMIZE(VAL(RIGHT$(TIMES$,2))):RETURN
```

Use the following lines for the APPLE II, APPLE II+, APPLE //e, and APPLE //c.

```
900 HOME:RETURN
910 VTAB VT:HTAB HT:RETURN
920 FOR WS=1 TO WT:RETURN
930 RD=INT(RX*RND(1)+1):RETURN
940 KY$=NU$:KY=PEEK(KZ):IF KY<128 THEN RETURN
950 KY$=CHR$(KY-128):POKE KW,0:RETURN
960 NU$=CHR$(0):SW=40:SH=24:KZ=-16384:KW=-16368:GOSUB 970:RETURN
970 RETURN
```


THE CHAINS OF LOKI



THE EVIL LOKI BELIEVES HE HAS FINALLY FOUND A SPELL THAT WILL STOP THOR FROM BOTHERING HIM AGAIN—THE **SPELL OF THE MYSTIC CHAINS!** IT IS A SPELL YOU CAN CREATE IN YOUR COMPUTER!
READ ON, TRUE BELIEVER, TO FIND OUT HOW!

Program 1

Chains of Loki

Input the following program in your computer and run it. Be sure to try the program several times using different numbers.

Load your 900 lines before typing this program.

```
100 REM CHAINS OF LOKI
110 GOSUB 960:GOSUB 900
120 M=INT(SW/2):D=M/2
130 INPUT "ENTER A NUMBER FROM 1 TO 5 ";N
140 FOR I=1 TO 10
150 FOR J=1 TO 4 STEP N/10
160 Z1=M-ABS(INT(SIN(J)*D+0.5))
170 Z2=M+ABS(INT(SIN(J)*D+0.5))
180 HT=Z1:VT=SH:GOSUB 910
190 PRINT "O";
200 HT=Z2:VT=SH:GOSUB 910
210 PRINT "O"
220 NEXT J
230 NEXT I
240 END
```




LOKI IS GONE! BUT THOR IS STILL TIED UP. THE CHAINS ARE TIGHTENING AROUND HIM. IF HE DOES NOT FIGURE OUT A WAY TO BREAK THEIR POWER, HE WILL BE STRANGLED TO DEATH! WAIT! LOKI'S SPELL BOOK IS OPEN! THOR TRIES TO REPEAT THE MAGIC WORDS HE READS. BUT IT'S NO USE. THOR CAN'T BREAK THE SPELL. BUT YOU CAN!

THE SPELL WILL ONLY WORK IF THE WORDS ARE PUT IN THE RIGHT ORDER—ALPHABETICAL ORDER. THOR DOESN'T HAVE TIME TO DO IT. THE CHAINS ARE TOO TIGHT. BUT YOU CAN DO IT FOR HIM WITH THE PROGRAM ON THE NEXT PAGE!

QUICK TURN THE PAGE!

The **MAGIC WORDS**



Program 2

Alpha Sort

Input the following program in your computer and run it. The jumbled message from Loki's spell book is already in the program as a data statement in Lines 420, 430, 440.

You can use this program to alphabetize any information that you might want to keep in alphabetical order. Just enter the information in any order using the data statement in Lines 420, 430, 440.

Or, you can use this program to send a coded message to a friend. Make up a message that makes sense if it is read in alphabetical order. Then mix up the order of the words and enter them in Lines 420, 430, 440. The last data statement must be DATA STOP.

Only a friend who has this program will understand how to decode your message.

Load your 900 lines before typing this program.

```
100 REM ALPHA SORT
110 DIM A$(25)
120 GOSUB 960:GOSUB 900
130 N=1
140 READ A$(N)
150 IF A$(N)="STOP" THEN N=N-1:GOTO 170
160 N=N+1:GOTO 140
170 FOR I=1 TO N-1
180 FOR J=I TO N
190 HT=1:VT=1:GOSUB 910
200 FOR II=1 TO N
210 IF II=I THEN P$=" ==> "
220 IF II=J THEN P$=" ==> "
230 VT=II:GOSUB 910
240 PRINT P$;:P$=" "
250 NEXT II
260 IF A$(I)<A$(J) THEN 300
270 T$=A$(I)
280 A$(I)=A$(J)
290 A$(J)=T$
300 HT=1:VT=1:GOSUB 910
310 WT=WU:GOSUB 920
320 FOR II=1 TO N
330 IF II=I THEN P$=" ==> "
340 PRINT P$;A$(II);" "
350 P$=" "
360 NEXT II
```



```
370 NEXT J,I
380 HT=1:VT=1:GOSUB 910
390 VT=N:HT=17:GOSUB 910
400 PRINT "*** DONE ***":END
410 REM ** PUT THE WORDS HERE **
420 DATA NOW,VANISH,GRIPPING,CHAINS
430 DATA RELEASE,YOUR,HOLD,THAT,BINDS,ALL
440 DATA STOP
```


ALPHA SORT
CHANGES FOR SPECIFIC COMPUTERS

FOR ALL COMPUTERS CHANGE LINE 970 AS FOLLOWS


```
970 WU=100:RETURN
```

FANTASTIC FOUR

LEAD BY **MR. FANTASTIC** (REED RICHARDS, ONE OF THE MOST BRILLIANT SCIENTIFIC MINDS OF OUR TIME), THE FANTASTIC FOUR HAS MET MANY FOES. BUT WILL **THE INVISIBLE GIRL**, SUE STORM RICHARDS (REED'S WIFE), **THE HUMAN TORCH** (JOHN STORM, SUE'S BROTHER), AND **THE THING** (BENJAMIN GRIMM) BE UP TO THE GREATEST COMPUTER CHALLENGE OF ALL TIME?



THE GREATEST COMPUTER CHALLENGE



Program 3

NIM

Input the following program in your computer and run it. The object of the game is to leave the opposing player with the last stick, represented on the computer by the number of asterisks.

You get to choose whether you or the computer will go first, and you also choose the number of sticks you want to play with.


Good luck! The computer has been programmed to make it tough for you.


Load your 900 lines before typing this program.

```
100 REM NIM
110 GOSUB 900
120 PRINT "WHO PLAYS FIRST?"
130 INPUT "0-YOU 1-ME";P
140 INPUT "HOW MANY STICKS?";S
150 GOSUB 380
160 ON P+1 GOTO 170,240
170 REM YOU CHOOSE
180 IF S=1 THEN PRINT "OH GOODY, I WIN!":GOTO 350
190 INPUT "HOW MANY FOR YOU?";T
200 IF T>3 OR T<1 OR T>S THEN PRINT "SORRY, TRY AGAIN.":GOTO 190
210 S=S-T
220 PRINT "YOU TOOK ";T;" STICKS.":PRINT "THAT LEAVES ";S
230 GOSUB 380
240 IF S<5 THEN 320
250 T=S-(INT(S/4)*4+1)
260 IF T=0 THEN T=3
270 IF T=-1 THEN T=1
280 S=S-T
290 PRINT "I TAKE ";T;" STICKS.":PRINT "THAT LEAVES ";S
300 GOSUB 380
310 GOTO 170
320 IF S=1 THEN PRINT "OH FOO! I LOST!":GOTO 350
330 T=S-1
```

```
340 GOTO 280
350 INPUT "WANT TO PLAY AGAIN? (Y/N)";Y$
360 IF Y$="Y" THEN 100
370 END
380 FOR I=1 TO S
390 PRINT "**";
400 NEXT I
410 PRINT:PRINT
420 RETURN
```







DOCTOR DOOM'S TIME MACHINE

DOCTOR DOOM INVENTED A TIME MACHINE
AND PLANS TO GO BACK IN TIME AND
**STEAL ALL THE KNOWLEDGE OF EARTH'S
GREATEST SCIENTISTS!** TURN THE PAGE TO
LEARN THE PROGRAM HE USED TO
ACTIVATE THE TIME MACHINE.

Program 4

Time Spirals

Input the following program in your computer and run it. Watch the screen and prepare for your dizzying trip back through time with Dr. Doom.

Interrupt the program when you feel you've traveled enough.

Load your 900 lines before typing this program.

```
100 REM TIME SPIRALS
110 GOSUB 900:GOSUB 960
120 CH$="*"
130 PW=1
140 XO=SW/2:YO=SH/2
150 FOR I=1 TO SH-2
160 IF PW=1 THEN PW=-1:GOTO 180
170 PW=1
180 P2=-PW
190 XN=XO-I*PW
200 FOR PX=XO TO XN STEP P2
210 VT=YO:HT=PX:GOSUB 910
220 PRINT CH$
230 NEXT
240 XO=XN
250 YN=YO-I*PW
260 FOR PY=YO TO YN STEP P2
270 VT=PY:HT=XO:GOSUB 910
280 PRINT CH$
290 NEXT
300 YO=YN
310 NEXT
320 IF CH$ <> " " THEN CH$=" ":GOTO 140
330 RX=CS:GOSUB 930
340 CH$=CHR$(RD+BC)
350 GOTO 140
```

TIME SPIRALS
CHANGES FOR SPECIFIC COMPUTERS

1. RADIO SHACK COLOR COMPUTER

970 BC=120:CS=78:RETURN

2. COMMODORE 64 AND VIC-20

970 BC=33:CS=95:RD=RND(-TI):RETURN

3. IBM PC AND IBM PCjr


970 BC=33:CS=95:RANDOMIZE VAL(RIGHT\$(TIME\$,2)):RETURN

4. ALL OTHERS (APPLE][, APPLE][+, APPLE //e, and APPLE //c, TRS 80
III)

970 BC=33:CS=95:RETURN

ROMAN DENUMERALIZER

THE FANTASTIC FOUR HAVE FOLLOWED DOOM INTO THE PAST. BUT WHAT YEAR ARE THEY IN? THE NUMBERS THEY SEE SPEEDING BY ARE IN ROMAN NUMERALS. LUCKILY REED HAS INVENTED A HANDY ROMAN DENUMERALIZER TO HELP THEM OUT.



MCMLXXXIV

MCCCLXXX

MDCCXXII



MDCCCXXXVI

CDXXXII

THE DENUMERALIZER
IS ON THE
NEXT PAGE!

Program 5

Roman Denumeralizer

Input the following program in your computer and run it. Run each of the Roman numerals from the previous page through the Denumeralizer Machine. After each Roman numeral has been completed, run the program again to do the others.

Load your 900 lines before typing this program.

```
100 REM ROMAN DENUMERALIZER
110 GOSUB 960:GOSUB 900
120 PRINT "ENTER YOUR ROMAN":PRINT "NUMERAL":PRINT
130 INPUT A$
140 GOSUB 900
150 PRINT "    ROMAN NUMERAL":PRINT "    CONVERTER MACHINE!"
160 VT=5:HT=11-INT(LEN(A$)/2):GOSUB 910:PRINT A$;
170 VT=7:HT=1:GOSUB 910
180 PRINT "          *---* "
190 PRINT "          !   ! "
200 PRINT " -----!   ! "
210 PRINT " *-----*"
220 PRINT " ! THE NUMBER IS !"
230 PRINT " !               !"
240 PRINT " !               !"
250 PRINT " *-----* "
260 FOR I=1 TO LEN(A$)+1
270 VT=8:HT=17:GOSUB 910:PRINT ".*.";
280 VT=9:HT=17:GOSUB 910:PRINT M1$;
290 VT=8
300 A1$="":IF LEN(A$)=I-1 THEN 320
310 A1$=LEFT$(A$,LEN(A$)-I+1)
320 HT=14-LEN(A$)+I:GOSUB 910:PRINT " ";A1$;
330 WT=WU:GOSUB 920
340 VT=8:HT=17:GOSUB 910:PRINT "O.O";
350 VT=9:HT=17:GOSUB 910:PRINT M2$;
360 WT=WU:GOSUB 920
370 IF I>LEN(A$) THEN 510
```

```

380 C$=MID$(A$,LEN(A$)-I+1,1)
390 IF C$="I" THEN N=1
400 IF C$="V" THEN N=5
410 IF C$="X" THEN N=10
420 IF C$="L" THEN N=50
430 IF C$="C" THEN N=100
440 IF C$="D" THEN N=500
450 IF C$="M" THEN N=1000
460 IF N=>0 THEN O=N:GOTO 480
470 T=T-N:GOTO 490
480 T=T+N
490 VT=13:HT=10:GOSUB 910:PRINT T;
500 NEXT I
510 VT=15:HT=1:GOSUB 910
520 END

```

ROMAN DENUMERALIZER
 CHANGES FOR SPECIFIC COMPUTERS

1. COMMODORE 64 and VIC-20

```

970 M$=CHR$(164):M1$=M$+M$+M$:M2$="("+M$+")":WU=200:RETURN

```

2. ALL OTHERS (APPLE][, APPLE][+, APPLE //e, and APPLE //c, TRS 80 III, TRS 80 COLOR COMPUTER, IBM PC, AND IBM PCjr)


```

970 M$=CHR$(95):M1$=M$+M$+M$:M2$="("+M$+")":WU=200:RETURN

```

THE MAZE OF DOOM





THE FANTASTIC FOUR HAS FOLLOWED DOCTOR DOOM INTO THE PAST. THEY ARE IN A CHAMBER DEEP INSIDE AN EGYPTIAN PYRAMID. **THEY MUST GET TO DOOM BEFORE HE DISCOVERS THE MYSTIC SECRETS OF THE PYRAMID.** BUT FIRST THEY MUST PASS THROUGH A MYSTERIOUS MAZE. YOU AND YOUR COMPUTER MUST HELP THEM. **BUT BE CAREFUL!** DOOM HAS BOOBY-TRAPPED THE MAZE. A WRONG TURN COULD MEAN DEATH!

Maze of Doom

Input the following program in your computer and run it. Try and find your way out of the Maze of Doom.

You may want to make a map of the maze with pencil and paper as you go through it. You can use page 34 to make your map.

There is a correct way out of the maze, so take as many tries at solving the maze as you want to. Good luck!

Load your 900 lines before typing this program.





```
100 REM MAZE OF DOOM
105 GOSUB 960:GOSUB 900:RESTORE
110 DATA 7,11,7,3
120 DATA ".....*....."
130 DATA ".@*****@.."
140 DATA ".....*.*.*.."
150 DATA "....@*****.."
160 DATA ".....**@"
170 DATA ".*****.."
180 DATA "...*....."
190 DATA YOUR WAY IS BLOCKED BY STEEP WALLS
200 DATA YOU WERE EATEN BY A MONSTER
210 DATA OK
220 READ N,N1,R,C
230 FOR I=1 TO N
240 READ M$(I)
250 NEXT I
260 READ W$(1),W$(2),W$(3)
270 PRINT "WHICH DIRECTION"
280 PRINT "1-NORTH 2-SOUTH":PRINT "3-EAST 4-WEST"
290 INPUT D
300 ON D GOSUB 370,380,400,390
310 IF (R=1)*(MID$(M$(1),C,1)="*") THEN PRINT " GOOD WORK":END
320 PRINT W$(W):IF W=2 THEN 340
330 GOTO 270
```


```

340 PRINT "TRY AGAIN (Y/N)":INPUT Y$
350 IF Y$="Y" THEN 100
360 END
370 MX=1:AD=-1:GOSUB 410:RETURN
380 MX=N:AD=1:GOSUB 410:RETURN
390 MX=1:AD=-1:GOSUB 450:RETURN
400 MX=N:AD=1:GOSUB 450:RETURN
410 IF R=MX THEN W=1:RETURN
420 IF MID$(M$(R+AD),C,1)="." THEN W=1:RETURN
430 IF MID$(M$(R+AD),C,1)="@" THEN W=2:RETURN
440 IF MID$(M$(R+AD),C,1)="*" THEN W=3:R=R+AD:RETURN
450 IF C=MX THEN W=1:RETURN
460 IF MID$(M$(R),C+AD,1)="." THEN W=1:RETURN
470 IF MID$(M$(R),C+AD,1)="@" THEN W=2:RETURN
480 IF MID$(M$(R),C+AD,1)="*" THEN W=3:C=C+AD:RETURN

```

MAZE OF DOOM WORK SHEET





THE DANGER OF THE CODED STONES

THE FANTASTIC FOUR ARE TRYING TO STOP DOCTOR DOOM! THEY ARE TRYING TO CATCH HIM IN THE EGYPTIAN CHAMBER WHEN THE CEILING STARTS TO FALL ON THEM. THE ONLY WAY THEY CAN AVOID BEING HIT IS BY DECODING THE LETTERS ON EACH CODED PIECE. REED INVENTED A PROGRAM TO HELP HIM. BUT IT'S UP TO YOU TO USE IT! IT'S ON THE NEXT PAGE!

Program 7

Scramble Drop

Input the following program in your computer and run it. Try to unscramble the coded words. Every time you miss, the dangerous scrambled stones come closer to crashing down on The Fantastic Four!

Load your 900 lines before typing this program.

```
100 REM SCRAMBLE DROP
110 GOSUB 960
120 N=10
130 C=SH-8:MS=INT(SW/2)
140 C1=C:RX=N:GOSUB 930:R1=RD
150 FOR I=1 TO N
160 READ A$(I)
170 NEXT I
180 C$=""
190 GOSUB 370
200 GOSUB 900
210 MD=INT(LEN(S$)/2)
220 VT=C:HT=MS-MD:GOSUB 910
230 PRINT C$
240 VT=SH-5:HT=1:GOSUB 910:PRINT "
250 VT=SH-5:HT=MS-MD:GOSUB 910
260 PRINT LEFT$("-----",LEN(S$))
270 VT=1:HT=1:GOSUB 910
280 INPUT "UNSCRAMBLE->";Z$
290 IF Z$<>S$ THEN C=C+1:IF C<SH-5 THEN 220
300 GOSUB 900
310 IF C=SH-5 THEN PRINT "YOU DIED AFTER ";KT;" WORDS":C=C1
320 IF Z$=S$ THEN PRINT "* YOU MADE IT ALIVE *"
330 PRINT
340 INPUT "WANT TO TRY AGAIN";Y$
350 IF Y$="Y" THEN KT=KT + 1 :GOTO 180
360 END
370 X$="":R1=R1+1:IF R1>N THEN R1=1
380 FOR XX=1 TO LEN(A$(R1))
390 X$=X$+CHR$(ASC(MID$(A$(R1),XX,1))-1)
400 NEXT XX
```

```

410 S$=X$
420 FOR K=LEN(X$) TO 1 STEP -1
430 RX=K:GOSUB 930:R=RD
440 C$=C$+MID$(X$,R,1)
450 IF K=1 THEN 500
460 IF R=K THEN X$=LEFT$(X$,R-1):GOTO 490
470 IF R=1 THEN X$=MID$(X$,2):GOTO 490
480 X$=LEFT$(X$,R-1)+MID$(X$,R+1)
490 NEXT K
500 RETURN
510 DATA NVNNZ,QZSBNJE,UPNC
520 DATA QBQZSVT,DMFPQBUSB
530 DATA TQIJOY,QIBSBPI
540 DATA EFTFSU,PBTJT,BTQ,DBNFM

```

THE CHALLENGE OF THE CENTURIES



DOCTOR DOOM HAS MADE HIMSELF PHARAOH OF EGYPT! THE ONLY WAY HE WILL LEAVE AND RETURN BACK TO THE 20TH CENTURY IS IF REED CAN BEAT HIM IN A SPECIAL NUMBERS MATCH GAME! BUT REED MUST MATCH THE NUMBERS FAST, BEFORE THE SAND IN THE HOURGLASS HAS COMPLETELY RUN TO THE BOTTOM! TURN THE PAGE TO SEE HOW TO PROGRAM YOUR COMPUTER FOR THE SAME CHALLENGE!



Program 8

Challenge Game

Input the following program and run it. This program is a computer version of the match game or concentration. The idea is to find all the pairs of letters that are hidden under the asterisks that will appear on the screen.

When you run the program, first the computer will ask you to select a skill level—1, 2, or 3. (Three is the hardest.) Select a skill level by hitting either 1, 2, or 3 and then return. Then study the grid in front of you. The easiest grid will look something like this:

	A	B
1	*	*
2	*	*

When the computer asks for your guess, you must tell it which two asterisks to remove by typing in the letter of the vertical row and then the number of the horizontal row the first asterisk. Then type in a comma, then the letter and number of the location of the second asterisk. You must type the letter first. So if you guess A1, A2, the computer would remove the asterisk in the top left-hand corner and the one directly under it. If they both show the same letter, you get a match. The computer will keep track of how many guesses it takes you to get all the matches. REMEMBER: USE CAPITAL LETTERS AND TYPE THE LETTER FIRST, THEN THE NUMBER.

Load your 900 lines before typing this program.

```
100 REM CHALLENGE GAME
110 GOSUB 960:GOSUB 900
120 DIM V(36),A1(36)
130 PRINT "SKILL LEVEL 1,2 OR 3?"
140 INPUT "(3 IS HARDEST) ->";N
150 N=N*2
160 GOSUB 330:GOSUB 180:GOSUB 450
170 END
180 GOSUB 900
190 PRINT SP$;" ";
200 FOR I=0 TO N-1:PRINT CHR$(ASC("A")+I);" ";:NEXT I:PRINT
210 FOR I=0 TO N-1
220 PRINT STR$(I+1);" ";
230 FOR J=1 TO N
240 L=I*N+J
250 C=A1(L)*ABS(V(L)=1)
260 C=C+ASC(" ")*ABS(V(L)=-1)+ASC("**")*ABS(V(L)=0)
270 PRINT CHR$(C);" ";
280 NEXT J
290 PRINT
300 NEXT I
310 PRINT
320 RETURN
330 REM SHUFFLE
340 FOR I=0 TO 1
350 A$=LEFT$("ABCDEFGHIJKLMNOPQRSTUVWXYZ",N*N)
```

```

360 FOR K=(N*N/2) TO 1 STEP -1
370 RX=K: GOSUB 930
380 A1(K+N*N/2*I)=ASC(MID$(A$,RD,1))
390 IF (K=1) THEN A$=LEFT$(A$,RD-1):GOTO 420
400 IF RD=1 THEN A$=MID$(A$,2):GOTO 420
410 A$=LEFT$(A$,RD-1)+MID$(A$,RD,1)
420 NEXT K
430 NEXT I
440 RETURN
450 REM GET GUESS
460 INPUT "YOUR GUESS ";G$,G1$
470 GS=GS+1
480 X=ASC(LEFT$(G$,1))-ASC("A")+1+N*(ASC(MID$(G$,2,1))-ASC("1"))
490 Y=ASC(LEFT$(G1$,1))-ASC("A")+1+N*(ASC(MID$(G1$,2,1))-ASC("1"))
500 IF (V(X)=-1)+(V(Y)=-1) THEN PRINT "TRY AGAIN!":GOTO 460
510 V(X)=1
520 V(Y)=1
530 GOSUB 180
540 IF A1(X)=A1(Y) THEN V(X)=-1:V(Y)=-1:SC=SC+1:GOTO 560
550 V(X)=0:V(Y)=0
560 IF SC=N*N/2 THEN PRINT "YOU GOT THEM ALL IN":PRINT GS;"
    GUESSES":END
570 INPUT "PRESS RETURN TO GO ON";Q$
580 GOSUB 180
590 GOTO 460

```

CHALLENGE GAME CHANGES FOR SPECIFIC COMPUTERS

1. COMMODORE 64 and VIC-20.

```
970 SP$=" ":RD=RND(-TI):RETURN
```

2. IBM PC and PCjr.

```
970 SP$=" ":RANDOMIZE VAL(RIGHT$(TIMES$,2)):RETURN
```

3. TRS 80 MODEL III and COLOR COMPUTER.

```
970 SP$=" ":RETURN
```

4. APPLE II

```
970 SP$="":RETURN
```

**NICE WORK,
GUYS!**

THE FANTASTIC FOUR (WITH YOUR HELP) HAS DEFEATED DR. DOOM THIS TIME. BUT ALL IS NOT QUIET IN THE MARVEL UNIVERSE. THE SUPER HEROES ARE BESIEGED ON ALL SIDES RIGHT AT THIS VERY MOMENT.


SPIDER-MAN IS HAVING TROUBLE WITH THE HOBGOBLIN. QUICK! HE NEEDS YOUR HELP!





DISARM THE BOMB!

SPIDER-MAN IS UP TO HIS EYES IN TROUBLE. THE HOBGOBLIN HAS LEFT A TICKING TIME BOMB IN THE EMPIRE STATE BUILDING. IT'S UP TO SPIDEY TO DISARM THE BOMB—OR HUNDREDS OF LIVES WILL BE LOST! SPIDEY KNOWS THE BOMB IS COMPUTER CONTROLLED. IN FACT HE'S GOT THE CONTROL PROGRAM RUNNING. THERE ARE 13 NUMBERS THAT ARE NEEDED TO TURN OFF THE COUNTDOWN. THE PROGRAM HAS BEEN RIGGED. IT WILL ONLY GIVE OUT ONE OF THE NUMBERS. CAN YOU HELP? IF YOU CAN MAKE THE COMPUTER GIVE YOU THE OTHER TWELVE NUMBERS, THERE MIGHT STILL BE TIME TO SAVE THE BIG APPLE!



Program 9

Disarm Bomb

Input the following program in your computer and run it. As you can see, the program will only give you the first number. Can you debug the program?

List the program and see if you can discover the problem.

If you need help, consult the Hint Section, page 89.

Load your 900 lines before typing this program.

```
100 REM DISARM BOMB
110 GOSUB 900
120 READ A$
130 PRINT "HERE ARE THE ";LEN(A$);" CODES:"
140 PRINT
150 FOR I=1 TO LEN(A$)
160 GOSUB 210
170 NEXT I
180 PRINT
190 PRINT "YOU DID IT!"
200 END
210 V=ASC(MID$(A$,I,1))
220 PRINT "CODE #";I;" IS ";V
230 END
240 DATA "XMENSPIDERMAN"
```


CAPTAIN AMERICA'S INFORMATION INPUT



COMP

INT

OFF/ON

*

....
....

I'M UPDATING ALL THE FILES OF **EVERY**
SUPER HERO WHO WAS AND IS AN AVENGER.
TURN THE PAGE TO FIND OUT WHAT PRO-
GRAM I'M USING, AND HOW YOU CAN USE IT
TO MAKE YOUR OWN PERSONAL DATA FILE.



Program 10

Phone Book

Input the following program in your computer and run it.

Answer "Y" to the first question, then pick a letter of the alphabet. The computer will give you a listing of names and phone numbers for that letter. Hint: try P or S first.

You can use the program to create a computer phone book. Add your friends' names and telephone numbers in the data statements from Line 600 to 899.

Be sure and list the program to see the correct form for inputting data. Then save the program.

Load your 900 lines before typing this program.

```
100 REM PHONE BOOK
110 DIM LA$(30),FR$(30),PN$(30),R(30)
120 GOSUB 960:GOSUB 900
130 PRINT "HAVE YOU ENTERED THE DATA?"
140 INPUT Y$
150 IF Y$<>"Y" THEN 570
160 FOR K=1 TO 30:R(K)=K:NEXT
170 N=1
180 READ LA$(N)
190 IF LA$(N)="LAST" THEN 230
200 READ FR$(N),PN$(N)
210 N=N+1
220 GOTO 180
230 GOSUB 900
240 N=N-1
250 FOR I=1 TO N-1
260 FOR J=I TO N
270 IF LA$(R(J))+FR$(R(J))=>LA$(R(I))+FR$(R(I)) THEN 310
280 T=R(J)
290 R(J)=R(I)
300 R(I)=T
310 NEXT J,I
320 GOSUB 900
330 INPUT "LETTER ";L$
340 GOSUB 900
350 L=1
360 FOR K=1 TO N
370 I=R(K)
380 IF L$=" " THEN 400
```

```

390 IF LEFT$(LA$(I),1)<>L$ THEN 450
400 PRINT LA$(I);", ";FR$(I)
410 DT$="....."
420 PRINT LEFT$(DT$,SW-13);PN$(I)
430 PRINT
440 L=L+1
450 IF 3*L<SH-2 THEN 510
460 L=1
470 VT=SH:HT=1:GOSUB 910
480 PRINT "PRESS A KEY FOR MORE.";
490 GOSUB 940:IF KY$=NU$THEN 490
500 GOSUB 900
510 NEXT K
520 VT=SH:HT=1:GOSUB 910
530 PRINT "Q)UIT  A)NOTHER";
540 GOSUB 940:IF KY$="" THEN 540
550 IF KY$="A" THEN 320
560 IF KY$<>"Q" THEN 540
570 END
600 REM ENTER THE DATA HERE
602 DATA SPIDER-MAN,,345-0000
604 DATA PARKER,PETER,123-4567
606 DATA STORM,SUE,111-1111
899 DATA LAST

```





DAREDEVIL

AND THE

DEATH METER

DAREDEVIL IS TRAPPED! THE ONLY WAY HE CAN ESCAPE IS BY STOPPING THE WALL AT THE EXACT METER READING THE COMPUTER PROGRAM IS LOOKING FOR. THE METER ON THE SCREEN WOULD HELP—BUT DAREDEVIL CAN'T SEE—SO HE CAN'T USE IT. YOU MUST WATCH THE METER AND GET HIM OUT OF THIS TIGHT SPOT!

Program 11

Death Meter

Input the following program in your computer and run it. Try and stop the death meter.

Five correct meter readings are needed to stop the closing door. Watch the asterisks climb the meter and stop it when you think it has reached the equivalent of the fraction on the screen. You can hit any key to stop the meter.

Load your 900 lines before typing this program.

```
100 REM DEATH METER
110 GOSUB 900
120 GOSUB 960
130 PRINT "  DEATH METER"
140 SY=SH-3: SX=SW-10
150 RX=9: GOSUB 930: N=RD
160 GOSUB 930: D=RD
170 IF (D<=1) THEN 150
180 IF N>=D THEN 150
190 FOR I=1 TO SY
200 VT=I: HT= SX+2: GOSUB 910
210 PRINT "==.=="
220 NEXT I
230 VT=INT(SY/4*3): HT=1: GOSUB 910
240 PRINT STR$(N); "/" ; STR$(D);
250 FOR I=SY TO 1 STEP -1
260 VT=I: HT= SX+4: GOSUB 910
270 PRINT "**";
280 Z=Z+1
290 FOR J=1 TO WK
300 GOSUB 940
310 IF KY$<>NU$ THEN 340
320 NEXT J
330 NEXT I
340 REM GOT A KEYPRESS
350 IF ABS(Z/SY-N/D)>0.1 THEN 390
360 VT=SH-2: HT=1: GOSUB 910
370 PRINT "HOORAY YOU GOT IT!";
```



```

380 END
390 VT=SY+1-INT((N/D)*SY+.5):HT=SX:GOSUB 910
400 PRINT "---->X<---";
410 VT=SH-2:HT=1:GOSUB 910
420 PRINT "TOO BAD. YOU MISSED";
430 END

```

DEATH METER CHANGES FOR SPECIFIC COMPUTERS

1. COMMODORE 64 AND VIC-20

```

970 WK=50:RD=RND(-TI):RETURN

```

2. IBM PC AND IBM PCjr

```

970 WK=100:RANDOMIZE VAL(RIGHT$(TIME$,2)):RETURN

```

3. ALL OTHERS (APPLE][, APPLE][+, APPLE //e, and APPLE //c, TRS 80 III, TRS 80 COLOR COMPUTER.)


```

970 WK=50:RETURN

```

IRON MAN'S SECRET MESSAGE SCRAMBLER

IRON MAN MUST CONTACT CAPTAIN AMERICA, WHO IS ON A DANGEROUS MISSION AGAINST HYDRA. IRON MAN DOESN'T WANT HYDRA TO GET THE MESSAGE, SO HE'S GOT TO SEND IT TO CAPTAIN AMERICA IN CODE. TURN THE PAGE AND YOU CAN ENCODE THE MESSAGE FOR HIM!



Program 12

Encoder

Input the following program in your computer and run it. Type in the secret message, all in caps, one line at a time. Type "STOP" when the entire message has been encoded.

Here is the message to encode:

***CAP! THERE HAS BEEN A CHANGE OF PLANS!
HYDRA HAS LEARNED OF YOUR MISSION!
THEY HAVE SET A TRAP FOR YOU!***

Load your 900 lines before typing this program.


```
100 REM ENCODER
110 GOSUB 900:GOSUB 960
120 DIM A$(25)
130 FOR I=1 TO 25:A$(I)="":NEXT I
140 MM$="":M$=""
150 PRINT "ENTER SECRET MESSAGE"
160 PRINT:PRINT
170 INPUT M$
180 IF M$="STOP" THEN END
190 FOR I=1 TO LEN(M$)
200 CH$=MID$(M$,I,1)
210 IF CH$=" " THEN CH$="*"
220 MM$=MM$+CH$
230 NEXT I
240 MM=INT(LEN(MM$)/5)
250 IF MM=LEN(MM$)/5 THEN 280
260 MM$=MM$+LEFT$("+-#$", (MM+1)*5-LEN(MM$))
270 MM=MM+1
280 PRINT:PRINT
290 FOR I=0 TO 4
300 FOR J=1 TO MM
```

```
310 A$(J)=A$(J)+MID$(MMS,J+I*MM,1)
320 NEXT J,I
330 PRINT "THE CODED MESSAGE:"
340 PRINT
350 FOR I=1 TO MM
360 PRINT A$(I); " ";
370 NEXT
380 PRINT:PRINT
390 PRINT:GOTO 130
```

S.H.I.E.L.D.'S SPECIAL MESSAGE DECODER

CAPTAIN AMERICA IS GETTING
IRON MAN'S MESSAGE. BUT HE
MUST FIRST RUN IT THROUGH
S.H.I.E.L.D.'S DECODER PRO-
GRAM BEFORE HE CAN READ IT.
HERE IS THE ENTIRE MESSAGE:

1. ICASR *R*I* WED0Y IAIN0 LTV*U LEEF. **RO+
2. T*AEN HCK*T EA*BR NNIAA **NCN YS*KC ONT*E UEHE.
3. I*IFU *IOTT WNNEE I**ES LPIN. LON*+ *S*M- BIFI@ ETIN#
4. GL*** OUFIM OCRRA DKOON **MN+



Program 13

Decoder

Input the following program in your computer and run it. Type in the coded message one line at a time, all in caps. Type "STOP" when the entire message has been decoded.

You can use both the "Encoder" and "Decoder" programs for your own messages as well.


Load your 900 lines before typing this program.

```
100 REM DECODER
110 GOSUB 900:GOSUB 960
120 MM$="":A$="":M$=""
130 PRINT "ENTER MESSAGE"
140 PRINT:PRINT
150 INPUT M$
160 IF M$="STOP" THEN END
170 FOR I=1 TO LEN(M$)
180 CH$=MID$(M$,I,1)
190 IF CH$=" " THEN 220
200 IF CH$="*" THEN CH$=" "
210 MM$=MM$+CH$
220 NEXT
230 MM=INT(LEN(MM$)/5)
240 FOR I=1 TO 5
250 FOR J=0 TO MM-1
260 A$=A$+MID$(MM$,I+J*5,1)
270 NEXT J,I
280 PRINT:PRINT
290 PRINT "DECODED MESSAGE:"
300 PRINT:PRINT A$
310 PRINT:PRINT:GOTO 120
```


THE LETTER LESSON

E
K

W




L



E

OF THE DANGER ROOM!

PROFESSOR X HAS CREATED A NEW TEST FOR THE X-MEN IN THE DANGER ROOM. THEY MUST STOP ALL THE LETTERS FROM HITTING THE FLOOR. ANY LETTER THEY MISS CAUSES THE FLOOR TO RISE ONE FOOT. TURN THE PAGE TO SEE HOW WELL YOU DO ON PROFESSOR X'S TEST! BUT BE FAST!



Program 14

Typeup

Input the following program in your computer and run it. As the letters drop from the top of the screen, try to hit that key before the letters fall to the bottom.

Watch out! Every time you miss, the floor rises one notch, which makes the letters fall to the bottom even faster.

Load your 900 lines before typing this program.

```
100 REM TYPEUP
110 GOSUB 960
120 UL$=LEFT$("-----",SW-1)
130 GOSUB 900
140 DATA "POORLY, I'M AFRAID.", "OK FOR A BEGINNER.", "VERY
    WELL!", "TERRIFIC!!"
150 A$=LEFT$("ABCDEFGHIJKLMNOPQRSTUVWXYZ0123456789",SW-1)
160 FOR I=1 TO 4
170 READ C$(I):NEXT I
180 FOR F=SH - 1 TO 3 STEP -1
190 VT=F:HT=1:GOSUB 910
200 PRINT UL$
210 MX=36: IF SW<=MX THEN MX=SW-1
220 FOR S=3 TO F
230 I=I+1 : IF I>MX THEN I=1
240 VT=1:HT=1:GOSUB 910
250 PRINT MID$(A$,I,MX);
260 PRINT MID$(A$,1,I-1);
270 IF DT THEN 320
280 RX=MX:GOSUB 930:D=RD:DT=1
290 D$=MID$(A$,D,1)
300 IF D<I THEN D=D+MX
310 D=D-I+1
320 VT=S-1:HT=D:GOSUB 910
330 PRINT " "
340 VT=S:HT=D:GOSUB 910
350 PRINT D$
360 GOSUB 940
```

```

370 WT=WU:GOSUB 920
380 IF KY$=D$ THEN GOSUB 440
390 NEXT S
400 GOSUB 900
410 NEXT F
420 GOSUB 520
430 END
440 SC=SC+1+(24-SH)/24
450 DT =0
460 VT=S:HT=D:GOSUB 910
470 PRINT " "
480 VT=SH-1:HT=1:GOSUB 910
490 PRINT "SCORE ";INT(SC);
500 S1=SC:IF S1>26 THEN S1=26
510 RETURN
520 PRINT "YOUR SCORE WAS ";INT(SC );"YOU DID ";C$(INT(S1/7+1))
530 END

```

TYPEUP CHANGES FOR SPECIFIC COMPUTERS

1. VIC-20

```
970 WU=30:RD=RND(-TI):RETURN
```

2. COLOR COMPUTER

```
970 WU=30:RETURN
```

3. COMMODORE 64


```
970 WU=1:RD=RND(-TI):RETURN
```

4. IBM PC and PCjr

```
970 WU=1:RANDOMIZE VAL(RIGHT$(TIMES$,2)):RETURN
```

5. ALL APPLES and TRS 80 MODEL III2

```
970 WU=1:RETURN
```



PROFESSOR X'S COMPUTER IMAGE PROGRAM



PROFESSOR X HAS JUST COMPLETED A PROGRAM THAT LETS HIM DRAW WITH HIS COMPUTER. HE'S USING IT TO MAKE AN IMAGE OF ALL THE X-MEN'S ENEMIES AND RECORD IT IN THE MEMORY BANK. TURN TO PAGE 65 TO SEE HOW YOU CAN SET UP YOUR OWN PROGRAM TO DRAW ON YOUR COMPUTER.

Program 15

Sketch Pad

Input the following program in your computer and run it.

Hit D to draw with asteriks, or hit C to draw with any character on the keyboard.

I will draw up.

M will draw down.

J will draw left.

K will draw right.

E erases and N allows you to move the cursor without drawing. HAVE FUN!



Load your 900 lines before typing this program.

```
100 REM SKETCH PAD
110 GOSUB 960:GOSUB 900
120 VT=SH-3:HT=1:GOSUB 910
130 PRINT "D)RAW    MOVE COMMANDS"
140 PRINT "N)ODRAW   I  J  K  M
150 PRINT "C)HANGE   E)RASE
160 C$(0)="*"
170 C$(1)=" "
180 VT=1:HT=1:GOSUB 910
190 IF DR=1 THEN PRINT "NODRAW ";
200 IF DR=0 THEN PRINT "DRAW  ";
210 IF D=0 THEN D=1:GOTO 230
220 D=0
230 VT=V:HT=H:GOSUB 910:PRINT C$(D);
240 GOSUB 940:IF KY$=NU$ THEN 350
250 VT=V:HT=H:GOSUB 910:PRINT C$(DR);
260 A$=KY$:IF A$="I" THEN V=V-1*ABS(V>2):GOTO 350
270 IF A$="M" THEN V=V+1*ABS(V<SH-4):GOTO 350
280 IF A$="J" THEN H=H-1*ABS(H>1):GOTO 350
290 IF A$="K" THEN H=H+1*ABS(H<SW-3):GOTO 350
300 IF A$="N" THEN DR=1:GOTO 350
310 IF A$="D" THEN DR=0:GOTO 350
320 IF A$="C" THEN 360
330 IF A$="E" THEN 100
340 GOTO 240
350 GOTO 180
360 VT=1:HT=1:GOSUB 910
370 PRINT "NEW CHARACTER=> ";
380 GOSUB 940:IF KY$=NU$ THEN 380
390 C$(0)=KY$
400 VT=1:HT=1:GOSUB 910
410 PRINT "                ";:GOTO 180
```

SKETCH PAD CHANGES FOR SPECIFIC COMPUTERS

1. ALL COMPUTERS

970 V=8:H=8:RETURN



THOR AND THE WELL OF MIMIR

THOR HAS GONE TO SEEK
THE FUTURE FROM THE
FIERY WILL OF MIMIR.

EACH
WORD
THAT RISES
WEAVES
A SHAPE IN
THE MYS-
TIC SMOKE.
TURN THE
PAGE AND
YOU CAN
WEAVE A
MAGICAL
SPELL OF
YOUR
OWN!



Program 16

Wordweaver

Input the following program in your computer and run it.

Can you see any pattern in the way the letters are transformed by the computer?

Load your 900 lines before typing this program.

```
100 REM WORD WEAVER
110 GOSUB 960:GOSUB 900
120 W=INT(SW/6)
130 INPUT "ENTER YOUR NAME: ";N$
140 GOSUB 900:HT=1:VT=24:GOSUB 910
150 FOR P=1 TO 4
160 FOR M=1 TO LEN(N$)
170 Y=ASC(MID$(N$,M,1))
180 GOSUB 220
190 NEXT M
200 NEXT P
210 END
220 X(0)=Y
230 N=128
240 FOR I=1 TO 7
250 N=N/2
260 X(I)=INT(Y/N)
270 Y=Y-X(I)*N
280 NEXT I
290 FOR K=1 TO 2
300 FOR J=1 TO W
310 FOR I=2 TO 7
320 IF X(I)=1 THEN PRINT " ";
330 IF X(I)=0 THEN PRINT "X";
340 NEXT I
350 NEXT J
360 PRINT "X"
370 NEXT K
380 RETURN
```





By the Moons of Munipoor,
By Shiggard's jagged spike,
By the Vipers of Valthor,
Let the powers of darkness strike!

DOCTOR STRANGE AND THE COMPUTER SPELL



DOCTOR STRANGE AND THE DREADED DORMAMMU ARE LOCKED IN A DEADLY MYSTIC BATTLE. **DOCTOR STRANGE NEEDS YOUR HELP.** HE HAS GIVEN YOUR COMPUTER THE POWER TO SCRAMBLE DORMAMMU'S SPOKEN SPELLS, BUT YOU MUST ACTIVATE THE COMPUTER. DORMAMMU'S WORDS ARE HERE. THE PROGRAM YOU NEED IS ON THE NEXT PAGE.

Program 17

Backwards Spell

Input the following program in your computer and run it. Then type in Dormammu's spell. Watch what happens.

Load your 900 lines before typing this program.

```
100 REM BACKWARDS
110 GOSUB 900
120 PRINT "    LLEPS SDRAWKCAB "
130 GOSUB 960
140 REM
150 HT=1:VT=6:GOSUB 910
160 I=1:LL=5
170 GOSUB 940:IF KY$=NU$ THEN 170
180 IF KY$="%" THEN END
190 L$=KY$
200 A=ASC(L$)
210 IF A<>DL THEN 250
220 L$=" "
230 I=I-1
240 GOTO 270
250 IF A=13 THEN 320
260 IF I=0 THEN I=1
270 VT=LL:HT=1:GOSUB 910
280 PRINT SPC(SW-I);L$;
290 IF I=SW THEN 320
300 I=I+1-ABS(A=DL)
310 GOTO 170
320 LL=LL+1:IF LL>SH THEN PRINT:LL=SH
330 I=0:GOTO 170
```


BACKWARDS SPELL
CHANGES FOR SPECIFIC COMPUTERS

1. COMMODORE 64 and VIC-20

970 DL=20:RETURN

2. ALL OTHERS (APPLE][, APPLE][+, APPLE //e, and APPLE //c, TRS 80
III, TRS 80 COLOR COMPUTER, IBM PC, AND IBM PCjr)

970 DL=8:RETURN



THE FLIGHT OF THE BAT

THE PROGRAM WORKED! BUT DR. STRANGE HAS BEEN HURT. THE ONLY WAY HE CAN ESCAPE IS TO TURN HIMSELF INTO A BAT AND FLY HOME.

UNFORTUNATELY HIS INJURY HAS AFFECTED HIS SENSE OF DIRECTION. YOU MUST GUIDE HIS WAY. QUICK! INPUT THE PROGRAM ON THE NEXT PAGE.

Program 18

Bat Fly

Input the following program in your computer and run it. Try to keep the bat from flying into the walls on the top, bottom, left, and right.

The flying commands will appear on the screen.

A score on 7 or below will get him home safely.

Load your 900 lines before typing this program.

```
100 REM BAT FLY
110 GOSUB 960:GOSUB 900
120 PRINT "FLY THE BAT!":PRINT
130 PRINT "1 LEFT, 2 UP,":PRINT:PRINT"3 DOWN, 4 RIGHT":PRINT
140 PRINT "PRESS A KEY TO START"
150 GOSUB 940:IF KY$=NU$ THEN 150
160 GOSUB 900
170 REM
180 HH=SW/2:VV=SH/2
190 GOSUB 940:AS=KY$:IF KY$=NU$ THEN 270
200 IF HM<>0 AND VM<>0 AND TJ=0 THEN TJ=1
210 ON VAL(AS) GOTO 230,240,250,260
220 GOTO 190
230 HM=-1:GOTO 270
240 VM=-1:GOTO 270
250 VM=1:GOTO 270
260 HM=1
270 VT=1:HT=1:GOSUB 910:PRINT ER,TJ,:TJ=TJ+ABS(TJ>0)
280 IF TJ>200 THEN 400
290 GOSUB 420:PRINT " ";
300 HH=HH+HM:VV=VV+VM
310 IF HT=SW-6 THEN GOSUB 390
320 IF HT=1 THEN GOSUB 390
330 IF VV=SH THEN GOSUB 390
340 IF VV=0 THEN GOSUB 390
350 GOSUB 420:PRINT BS(I);
360 IF I=1 THEN I=2:GOTO 380
370 I=1
380 GOTO 190
390 HH=SW/2-3:VV=SH/2:ER=ER+1:RETURN
400 GOSUB 900:PRINT ER;" HITS";
410 END
420 VT=VV:HT=HH:GOSUB 910:RETURN
```

BAT FLY
CHANGES FOR SPECIFIC COMPUTERS

1. FOR THE COMMODORE 64 AND VIC-20



```
970 B$(1)=" "+CHR$(205)+"OO"+CHR$(206)+" "  
980 B$(2)=" "+CHR$(175)+"OO"+CHR$(175)+" "  
990 RETURN
```

2. ALL OTHERS (APPLE][, APPLE][+, APPLE //e, and APPLE //c, TRS 80
III, TRS 80 COLOR COMPUTER, IBM PC, AND IBM PCjr)

```
970 B$(1)=" "+CHR$(92)+"OO"+CHR$(47)+" "  
980 B$(2)=" "+CHR$(95)+"OO"+CHR$(95)+" "  
990 RETURN
```

FLASH BY

OH NO! SPIDER-MAN IS TRAPPED IN ONE OF DR. OCTOPUS' TRAPS. THERE IS A WAY OUT. HE NEEDS THE PASSWORD TO OPEN THE COMPUTERIZED LOCK. THE LETTERS OF THE PASSWORD ARE FLASHING BEFORE HIM—BUT THEY'RE GOING BY MUCH TOO FAST! YOU'VE GOT TO HELP SPIDER-MAN. RUN THE PROGRAM, THEN SEE IF YOU CAN FIGURE OUT A WAY TO SLOW IT DOWN SO YOU CAN READ THE PASSWORD.



Program 19

Flash By

Input the following program in your computer and run it. As you can see, it's moving much too fast to read.

Load your 900 lines before typing this program.

```
100 REM FLASHBY
110 GOSUB 900
120 GOSUB 960
130 READ C$
140 FOR I=1 TO LEN(C$)
150 A$=MID$(C$,I,1)
160 IF ((ASC(A$)>=ASC("A"))*(ASC(A$)<=ASC("Z"))) THEN 180
170 B$=B$+A$:GOTO 190
180 B$=B$+CHR$(ASC(A$)-1)
190 NEXT I
200 VT=INT(SH/2):HT=INT(SW/2)
210 FOR I=1 TO LEN(B$)
220 GOSUB 910
230 PRINT MID$(B$,I,1);
240 WT=1:GOSUB 920
250 NEXT I
260 END
270 DATA "UFSSJGJD UFOUBDMFT"
```


List the program and see if you can figure out a way to slow down the program.

Hint: How about a wait loop? Can you find it in the program?

Modify the program to slow it down.


Give up? See page 92 of the Hint Section for help.

Once you have slowed down the program,
write each letter of the password in the spaces below.



**POOR PETER PARKER! AS IF IT
ISN'T TOUGH ENOUGH FIGHTING
FOR THE CAUSE OF JUSTICE ALL
DAY AS SPIDER-MAN—NOW HE
HAS TO DO HIS INCOME TAX! AND
HIS CALCULATOR IS BROKEN!
RUN THE PROGRAM ON THE
NEXT PAGE AND SEE IF YOU CAN
TRAP THE BUG THAT'S MAKING
THE WEB-SLINGER'S CALCU-
LATOR GO NUTS!**

THE BROKEN CALCULATOR



Program 20

Broken Calculator

Input the following program in your computer and run it.

Give the calculator any two numbers when asked, and tell it whether you want to add, subtract, multiply or divide (+, -, x, /).

Whoops! Something's wrong. You're getting some right answers and some wrong ones. List the program and see if you can figure out what needs to be added or changed.

If you're stumped, consult the Hint Section, page 93.


Load your 900 lines before typing this program.

```
100 REM CALCULATOR
110 GOSUB 900
120 OP$=""
130 PRINT:INPUT "FIRST NUMBER";N1
140 PRINT:INPUT "OPERATION";OP$
150 IF OP$="" THEN END
160 PRINT:INPUT "SECOND NUMBER";N2
170 IF OP$="+" THEN 220
180 IF OP$="-" THEN 240
190 IF OP$="X" THEN 250
200 IF OP$="/" THEN 260
210 GOTO 110
220 PRINT:PRINT:PRINT N1;OP$;N2;" = ";N1+N2
230 GOTO 120
240 PRINT:PRINT:PRINT N1;OP$;N2;" = ";N1-N2
250 PRINT:PRINT:PRINT N1;OP$;N2;" = ";N1*N2
260 PRINT:PRINT:PRINT N1;OP$;N2;" = ";N1/N2
270 END
```




SUPER HINT SECTION

General Hints



1. As you know, the programs in this book won't run correctly if they aren't typed into your computer exactly as they are in the listings. Nearly everyone makes one or two typing mistakes, so don't be surprised if your program doesn't run as it should the first time.

To check for mistakes, simply LIST the program. For a short program, just type LIST and hit RETURN (or ENTER or whatever your computer calls the key that enters commands). On longer programs, it's best to LIST a few lines at a time, so the program doesn't scroll up the screen and go past you. To do this, simply type something like LIST 100-170 and only those lines will appear on the screen. Then after you've checked those, you can check the next few lines—until you find your mistake.

2. When you are finished with one program and about to start another, it's a good idea to type NEW. That will clear out any leftovers from the previous program that could mess up your new program.


3. Remember to use all capital letters to type in the programs or to input data. Some computers have a CAP LOCK key (or LOCK). If you press it, all the letters will appear in caps.

4. Some of these programs will run on forever unless you interrupt or break them.

All computers have a way to interrupt a program. On the Commodore, you press the RUN/STOP key. On the IBM it's two keys: the CTL and BREAK. You hold them down at the same time. On the APPLE, you hold down the CONTROL key and press the letter C. The Radio Shacks have a BREAK key. If your computer keyboard is different, just consult your owner's manual.

Note: You may be able to think of different ways to achieve the same effects as these programs. Often these programs were not written in the fastest or easiest way. That's because the programs were designed to run on several computers. Besides no two programmers ever write programs exactly the same way. If you want to customize or change these programs, go right ahead. That's part of the fun of computers! Good luck.

**TURN THE PAGE
FOR EVEN MORE
HELPFUL HINTS!**



Hints about Specific Programs

Program 1 **Chains of Loki**

Chains of Loki is a very simple plotting program. It plots a sine curve, and its mirror image. The number you are asked to input is like a radius, determining how stretched out the 'links' of the chain will be. A value of two seems to work about the best. Just for fun try 1.5 or 2.7 and see what happens.

Program 2 **Alpha Sort**

Alpha Sort is an alphabetic sorter program. It shows you visually how a computer sorts data using a method called a bubble sort. This is a very simple type of sort. It compares each word to the rest of the words and switches the places of the words whenever the top word is higher in the alphabet than the bottom word. It's called a bubble sort because words 'bubble-up' from the bottom.

Program 3 **NIM**

NIM is a computer version of a classic game where you have a bunch of sticks and you try to leave the 'other guy' with the last stick. You may take one, two or three sticks on your turn. How can you win? Here's a hint: If there are five sticks left, and your opponent must play, no matter how many he takes, you can leave him with one stick when you play next. If you can leave him with nine or thirteen or seventeen sticks, you can always make your oppo-

nent lose. Do you notice that four is added to each number (5, 9, 13, 17)? If there are 17 can you leave him with 13? If there are thirteen, can you leave him with nine? Do you see a pattern?

You can always win if you start with 17 sticks and make the computer go first. The strategy is simple. Whatever number of sticks the computer takes, you subtract that number from four. Then you take that number of sticks. So if the computer takes three, you take one. If the computer takes two, you take two and so on. It works every time. What if you go first? Hint: The computer knows the strategy. Maybe you should start with a different number of sticks. Try it and see what happens.

Program 4

Time Spiral

Time Spiral is a pretty and simple graphics program. It generates a square spiral with characters. Which characters does it build with? Look at line 970. BC is the ASCII value of the first character with which the spiral can be built. CS is the number of characters past BC that can be used. The ASCII value of the last available character is BC + CS. Try changing the values of CS and BC in your line 970 and see what happens. Be sure to keep the values above 32 or strange things may happen.

Program 5

Roman Denumeralizer

Roman Denumeralizer is a conversion program. You enter numbers in Roman numeral form (like MCMXLVII) and the machine munches and crunches them into our more familiar base 10 numbering system. Try some fun numbers like MMCCII. Do you know what it would be without running the program? Try to get 1985 to come out of the machine. Can you do it?

Program 6

Maze of Doom

This is a maze-generation program. The maze is designed in data statements at the beginning of the program. You can change the program and create your own maze if you follow certain rules. Here they are:

1. The path is identified by asterisks (*).
2. Walls are identified by periods (.).
3. Monsters are identified by at-signs (@).
4. The maze may be up to ten rows long and 80 wide.
5. The data statement at line 110 has this information:
 - a. The number of rows in the maze
 - b. The number of columns in the maze
 - c. The starting position row
 - d. The starting position column
6. You must start at the bottom of the maze and exit at the top.
7. There must be at least one complete path through the maze.

Program 7

Scramble Drop

Scramble Drop is a game where you try to unscramble words before they hit the ground. To make the game more fair, the scrambled words are coded by 'adding one' to each letter before it is used in the data statements. Also the variable N in line 120 is set to the number of words in the list (data statements). You can add words if you like. For example, to add CAT to the list, add one to N at line 120, then put DBU (C → D A → B T → U) in the Data statements at line 700 or higher. This way the words are scrambled when you input them. Then the program rearranges the letters to scramble them even more!

Program 8

Challenge

Challenge is a variation on the match game or concentration. It doesn't have to match only letters. You can change the program. If you want to match stars, numbers, dollar signs, you can. Just change the characters between the quotes at line 350. Be sure not to repeat any characters in the string.

Program 9

Disarm Bomb!

Disarm Bomb will help you save the day if you can make it work! There is a real problem in the program. Can you find it? Remember that every GOSUB needs a RETURN to get back. Look at line 160. There is a GOSUB to line 210. Do you see any RETURNs right after line 210? Oops! There's an END instead at line 230. Perhaps it should be something else? (Hint: It should tell the program to return from the subroutine.)

That wasn't hard now, was it?

Program 10

Phone Book

Phone Book is a phone book you can use! At line 600 there are data statements. You can put as many as thirty names in the data statements, and the phone book will list them for you in alphabetic order. You enter data with Last Name, First Name, Phone Number. Be sure to put all three things for each person. If, like Spider-Man, there is no first name, just use the comma after nothing. That's a 'null' entry. What if you have more than 30 names? Be sure to change the DIM statements in line 110 to dimension the strings for at least as many entries as you will have. Line 899 has a data statement with just the word LAST. That data statement must be there so the program knows when to stop.

Program 11

Death Meter

Death Meter challenges you to guess the fraction it's looking for or die trying. It's just a fun game. Notice how it uses the KEYPRESS subroutine instead of an INPUT statement to know when to stop the meter. Do you know why? If we tried to use an input statement, the computer would stop and wait for us to type a character and a return or enter. It would not do anything else except wait! The meter couldn't keep changing while we were waiting. We use KEYPRESS procedure so that we can just keep checking back at the keyboard to see if a key has been pressed, but also do other things in the meantime.

Program 12

Encoder

Encoder is a scrambler. It takes the letters of the words you type and shuffles them around in a special way. All the letters are still there, but they're not together as words anymore. Look closely at the scrambled message. Remember what you typed in? Now look at about every fifth letter . . . look familiar? What happened to the spaces? What character did they change to? Try to change the program so that they change to a different symbol. When you make up your own messages to encode, don't use commas or colons.

Program 13

Decoder

Docoder will undo what encoder does. If you changed the character that the space changed to in the encoder program, be sure to change it here, too, so that the programs match up.

Program 14

Typeup

Typeup is an example of a challenging game written in a few lines of BASIC code. You try to type letters before they hit the floor. This program will even help you to learn to type without looking at the keys! You may want to slow the program down a little. Set the value of WU in line 970 to a larger number to make the letters fall more slowly. Then speed it up as you get better at it.

Program 15

Sketch Pad

Sketch Pad is a fun sort of drawing program. Just because all these programs need to run on many microcomputers easily, we draw with letters instead of graphics. Use your imagination. Have fun! Try using many different characters to draw with. Which draw the best up and down? Which sideways? Which are best to fill in areas? Be sure to try ! = and #.

These work very well. Remember:

I will draw UP

M will draw DOWN

J will draw LEFT

K will draw RIGHT

Program 16

Wordweaver

Wordweaver isn't just another random pattern generator. The name or word you enter into the program determines what pattern you will generate. It's kind of like a binary 'Coat-of-Arms'. Try some patterns, like JO,JOE,JOANNE, and see how the patterns change. The actual picture that scrolls by is directly tied to the ASCII value of the characters in the name. Do you want to make the pattern with something other than 'X'? Try changing the character in quotes at line 330.

Program 17

Backwards Spell

Backwards Spell spells backwards! Type away and see the words spelled backwards on your screen! How is it done? Usually your computer types forwards only, but you can fool it with a program. Instead of letting the computer position the cursor where it wants, you tell your micro where to put the cursor—and you do it after every single letter you type in! That's all done when there is a call to the subroutine at line 910. Each time, we have to tell the subroutine both the horizontal and vertical position of the cursor, so that the next letter we print is to the left, not the right, of the last letter we printed.

Program 18

Bat Fly

Bat Fly is another fun game that does simple animation. The bat 'flies' by changing its form from /00/ to -00- as it moves. This is a common kind of computer animation. It fools your eyes into thinking that something is really moving.

Program 19

Flash By


FlashBy is another debugging program. The trick here is to slow down the program enough to read the hidden message. Remember that we are using line 920 as our WAIT LOOP in these programs. Where do we call that subroutine? What is the loop value (WT) set to? Would the program run faster or slower if WT was a larger number? Try it and see.


Program 20

Broken Calculator

Broken Calculator is a challenge to your programming skills. It is an example of a common programming error called 'falling-through'. You intend to execute a few lines of code, then skip over some others. Instead you do both the code you intended AND the lines that follow it. This is usually a problem when you have a series of IF-THEN statements. Your brain thinks one thing and your program really says another! Does that happen in this program? What happens at the IF-THEN statements? Where does the program go? Does it stop at JUST the line that you GOTO? How could you tell the program to GO back TO line 110 after printing a correct answer? Is one correction enough? Not quite. To fix the program, you must change the following lines:

```
245 GOTO 120
255 GOTO 120
265 GOTO 120
```





SUBJECT INDEX TO TYPES OF PROGRAMS

CODES:

Alpha Sort,	Page 16
Backwards Spell,	Page 72
Decoder,	Page 59
Encoder,	Page 56
Scramble Drop,	Page 36

DATA PROGRAMS:

Alpha Sort,	Page 16
Phone Book,	Page 48

DEBUGGING PUZZLES:

Broken Calculator,	Page 81
Disarm Bomb,	Page 44
Flash By,	Page 78

GAMES:

Bat Fly,	Page 75
Challenge Game,	Page 40
Death Meter,	Page 52
Maze of Doom	Page 32
NIM,	Page 20
Scramble Drop,	Page 36
Typeup,	Page 62

GRAPHICS:

Sketch Pad,	Page 65
-------------	---------

NUMBER GAMES:

Roman Denumeralizer,	Page 28
----------------------	---------

SPECIAL EFFECTS:

Backwards Spell,	Page 72
Chains of Loki,	Page 13
Time Spirals,	Page 24
Wordweaver,	Page 68

About the Authors

Richard Guaraldo is an electrical engineering graduate from Drexel University in Philadelphia, PA, and an avid microcomputer enthusiast. His love of micros began when he purchased his first APPLE II in 1980. He quickly became involved with the Maryland Crab Apples, a local APPLE users group, where he served as newsletter editor, secretary, and president. His interests have expanded to CP/M, the COMMODORE 64, and the IBM-PC. In addition to this book, he has written several educational packages for the COMMODORE 64 that are used in the Anne Arundel County, Maryland, Public School System. He resides in Crofton, Maryland, with his wife and daughters.

Susan M. Zakar is a computer systems analyst/programmer living in Gambrills, Maryland. She has a bachelor's degree in foreign languages from Western Washington State College. In 1978, she purchased her first personal computer (an Apple II) and became so enchanted with the world of microcomputers that she eventually made it her career. She has programmed on many different kinds of microcomputers, in FORTRAN, BASIC, Pascal, and assembly language. She has contributed to, and acted as editor to several microcomputer user group newsletters. She also chairs a microcomputer user group at work, and she has created programs for the successful computer/adventure paperback series *Micro Adventures™*. Her interest in programming for young people was inspired by her two-year-old, for whom she hopes to write programs which 'fascinate and entertain.'


Your computer will turn you into a Super Hero!

That's right! Your computer makes you part of the Marvel Super Heroes'™ world of adventure. Program your computer to help the Marvel Super Heroes defeat the forces of evil. **20 super programs make the challenge fun!**



Program your computer to guide The Fantastic Four through Doctor Doom's amazingly mysterious maze.

Save Daredevil from a tight squeeze! Only your computer can save a moving wall from crushing him.



Iron Man must send a secret message to Captain America. A super computer code lets you do it.

Help Mr. Fantastic play a challenging strategy game against the computer!



So turn on your computer and get ready to load it with codes, computer puzzles, games, and graphics!

PROGRAMS CAN BE RUN ON ALL THESE PERSONAL COMPUTERS!

**APPLE II • APPLE IIe • APPLE II+ • APPLE IIc • COMMODORE 64
VIC-20 • IBM PC • IBM PC JR. • RADIO SHACK COLOR COMPUTER
RADIO SHACK TRS-80™ MOD 3**

ISBN: 0-917657-05-5



DONALD I. FINE, INC.